

## More Practice with Main Effects and Interactions in General Linear Models

The models for this example come from Hoffman (in preparation) chapter 2. We will be examining the extent to which cognition (as measured by an information test outcome) can be predicted from age (centered at 85 years) grip strength (centered at 9 pounds), sex (with men as the reference group) and dementia status (none, future, current, with none as the reference) in a sample of 550 older adults.

### SAS Syntax and Output for Data Manipulation:

```
* Location for SAS files for these models - change this path to run on your computer;
%LET filesave= F:\Example Data\Chapter 2 Data;
LIBNAME filesave "&filesave.";

TITLE "Descriptive Statistics for Example Variables in Original Data";
PROC FREQ DATA=filesave.example2; TABLE sexMW demgroup; RUN;
PROC MEANS DATA=filesave.example2; VAR cognition age grip; RUN;
TITLE;
```

sexmw: Sex (0=Men, 1=Women)				
sexmw	Frequency	Percent	Cumulative Frequency	Cumulative Percent
0	227	41.27	227	41.27
1	323	58.73	550	100.00

demgroup: Dementia Diagnosis (1=None, 2=Future, 3=Current)				
demgroup	Frequency	Percent	Cumulative Frequency	Cumulative Percent
1	399	72.55	399	72.55
2	109	19.82	508	92.36
3	42	7.64	550	100.00

Variable	Label	N	Mean	Std Dev	Minimum	Maximum
cognition	cognition: Cognitive Outcome	550	24.8218182	10.9890273	0	44.0000000
age	age: Age in Years	550	84.9267863	3.4300288	80.0164932	96.9672832
grip	grip: Grip Strength in Pounds	550	9.1127273	2.9829543	0	19.0000000

```
* Import example data into work library and center predictors;
DATA work.example2; SET filesave.example2;
* Center continuous predictors;
age85 = age - 85; LABEL age85= "age85: Age in Years (0=85)";
grip9 = grip - 9; LABEL grip9= "grip9: Grip Strength in Pounds (0=9)";
* Creating manual contrasts between groups (demNF, demNC) to map onto model equation;
IF demgroup=1 THEN DO; demNF=0; demNC=0; END; * None;
ELSE IF demgroup=2 THEN DO; demNF=1; demNC=0; END; * Future;
ELSE IF demgroup=3 THEN DO; demNF=0; demNC=1; END; * Current;
ELSE DO; demNF=.; demNC=.; END; * Otherwise is missing;
LABEL demNF= "demNF: None vs. Future Dementia"
demNC= "demNC: None vs. Current Dementia";
RUN;
```

The accompanying syntax file then has three macro programs I wrote that do the following:

- R2Change**: calculates an F-test for the improvement in  $R^2$  from nested models
- Asymp**: calculates the asymptotic covariance matrix of the fixed effects needed for...
- Regions**: calculates regions of significance for continuous predictor interactions

**SAS PROC GLM Syntax and Output for Equation 2.7, in which the *i* subscript indicates person:**

$$\text{Cognition}_i = \beta_0 + \beta_1 (\text{Age}_i - 85) + \beta_2 (\text{Grip}_i - 9) + \beta_3 (\text{SexMW}_i) + e_i$$

```
TITLE1 "Eq 2.7: Model with Age, Grip, Sex (0=Men, 1=Women)";
PROC GLM DATA=work.example2;
* Model Y = fixed effects, options after / print extra output;
  MODEL cognition = age85 grip9 sexMW / SOLUTION SS3 EFFECTSIZE;
* Save pieces of output for use in macros (optional);
  ODS OUTPUT FitStatistics=work.Rsq2_7;
* Intercepts per sex (hold age=85, grip=9);
  ESTIMATE "Intercept for Men"   intercept 1 sexMW 0;
  ESTIMATE "Intercept for Women" intercept 1 sexMW 1;
* Need both RUN and QUIT, shut off title;
RUN; QUIT; TITLE1;
```

The EFFECTSIZE option provides additional output related to  $\eta^2$  and  $\omega^2$  (not shown to save space). The ODS OUTPUT is saving the  $R^2$  for use later.

**ANOVA Test of Significance for Model  $R^2$**

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	3	6574.67775	2191.55925	20.04	<.0001
Error	546	59721.86043	109.38070		
Corrected Total	549	66296.53818			

**Model  $R^2$**

R-Square	Coeff Var	Root MSE	cognition Mean
0.099171	42.13439	10.45852	24.82182

**ANOVA Table of F-tests for Fixed Effects (*won't be uniquely useful, so is omitted from now on*)**

Source	DF	Type III SS	Mean Square	F Value	Pr > F
age85	1	1172.922323	1172.922323	10.72	0.0011
grip9	1	1179.415578	1179.415578	10.78	0.0011
sexmw	1	1609.324132	1609.324132	14.71	0.0001

**ESTIMATE Statement Results: Model-Implied Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Intercept for Men	26.9594307	0.73887292	36.49	<.0001
Intercept for Women	23.1606507	0.60034925	38.58	<.0001

**Solution for Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Intercept	26.95943069	0.73887292	36.49	<.0001
age85	-0.43377191	0.13246378	-3.27	0.0011
grip9	0.54600194	0.16627659	3.28	0.0011
sexmw	-3.79877994	0.99035915	-3.84	0.0001

**Interpret each fixed effect:**

Intercept  $\beta_0 =$

Main effect of Age  $\beta_1 =$

Main effect of Grip Strength  $\beta_2 =$

Main effect of Sex  $\beta_3 =$

The intercept for women is calculated from:

**SAS PROC GLM Syntax and Output for Equation 2.8, adding 2 contrasts for dementia group:**

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Age}_i - 85) + \beta_2(\text{Grip}_i - 9) + \beta_3(\text{SexMW}_i) + \beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + e_i$$

We can use the model equation to calculate the **dementia group means** for predicted cognition:

$$\begin{aligned} \text{Cognition for None} &= \beta_0 \\ \text{Cognition for Future} &= \beta_0 + \beta_4 \\ \text{Cognition for Current} &= \beta_0 + \beta_5 \end{aligned}$$

We can determine the **differences between the dementia group means** as follows:

$$\begin{aligned} \text{None vs. Future} &= \text{Future} - \text{None} = (\beta_0 + \beta_4) - (\beta_0) = \beta_4 \\ \text{None vs. Current} &= \text{Current} - \text{None} = (\beta_0 + \beta_5) - (\beta_0) = \beta_5 \\ \text{Future vs. Current} &= \text{Current} - \text{Future} = (\beta_0 + \beta_5) - (\beta_0 + \beta_4) = \beta_5 - \beta_4 = -\beta_4 + \beta_5 \end{aligned}$$

**These values are then requested via the ESTIMATE statements below...**

```
TITLE1 "Eq 2.8: Model adding Dementia Group (demNF, demNC) as manual contrasts";
PROC GLM DATA=work.example2;
* Model Y = fixed effects, options after / print extra output;
MODEL cognition = age85 grip9 sexMW demNF demNC / SOLUTION SS3;
* Request group means (hold age=85, grip=9, men);
ESTIMATE "Intercept for None" intercept 1 demNF 0 demNC 0;
ESTIMATE "Intercept for Future" intercept 1 demNF 1 demNC 0;
ESTIMATE "Intercept for Current" intercept 1 demNF 0 demNC 1;
* Request group differences;
ESTIMATE "None vs. Future" demNF 1 demNC 0;
ESTIMATE "None vs. Current" demNF 0 demNC 1;
ESTIMATE "Future vs. Current" demNF -1 demNC 1;
* Save pieces of output for use in macros (optional);
ODS OUTPUT FitStatistics=work.Rsq2_8;
* Need both RUN and QUIT, shut off titles;
RUN; QUIT; TITLE1;
```

**ANOVA Test for Model R<sup>2</sup>**

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	5	18385.97930	3677.19586	41.75	<.0001
Error	544	47910.55888	88.07088		
Corrected Total	549	66296.53818			

R-Square	Coeff Var	Root MSE	cognition Mean
0.277329	37.80790	9.384609	24.82182

**ESTIMATES: Requested Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Intercept for None	29.2643254	0.69850792	41.90	<.0001
Intercept for Future	23.5423544	1.07852978	21.83	<.0001
Intercept for Current	12.7845121	1.53019333	8.35	<.0001
None vs. Future	-5.7219710	1.01907848	-5.61	<.0001
None vs. Current	-16.4798133	1.52275357	-10.82	<.0001
Future vs. Current	-10.7578423	1.70795708	-6.30	<.0001

**Solution for Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Intercept	29.26432541	0.69850792	41.90	<.0001
age85	-0.40573396	0.11889717	-3.41	0.0007
grip9	0.60422556	0.14977568	4.03	<.0001
sexmw	-3.65737421	0.89143262	-4.10	<.0001
demNF	-5.72197100	1.01907848	-5.61	<.0001
demNC	-16.47981327	1.52275357	-10.82	<.0001

I wrote a macro program to calculate the **R<sup>2</sup> change F-test** from adding these 2 fixed effects for dementia group (i.e., a df=2 test rather than for each separate contrast), which is called below:

```

/* To use R2Change Macro, enter:
   N=                Total Sample Size
   DataL=            Name of dataset that stores R2 for larger model (on ODS)
   NpredL=           Number of predictors in larger model
   DataS=            Name of dataset that stores R2 for smaller model (on ODS)
   NpredS=           Number of predictors in smaller model */
* Call macro to test R2 change from Eq 2_8 vs. Eq 2_7;
%R2Change(N=550, DataL=work.Rsq2_8, NpredL=5, DataS=work.Rsq2_7, NpredS=3);

```

Which then generates this output, indicating that the overall df=2 effect of group is significant

R2 Change F-Test for work.Rsq2_8 (L) vs. work.Rsq2_7 (S)								
Rsquare	Rsquare	Rsquare	Npred	Npred			Fvalue	Pvalue
L	S	Diff	L	S	DFnum	DFdenom		
0.27733	0.099171	0.17816	5	3	2	544	67.0557	0

**SAS PROC GLM Syntax and Output for Equation 2.8 using CLASS statement for sex and dementia groups instead of manual contrasts:**

```

* Create formats for output using categorical predictors;
* Then uses numeric/alphabetical order to order categories the way I want!;
PROC FORMAT;
  VALUE Gender 0="2Men" 1="1Women";           * So Men stay ref;
  VALUE Dementia 1="4None" 2="2Future" 3="3Current"; * So None stays ref;
RUN;

TITLE1 "Eq 2.8: Model adding Dementia Group (2=Future, 3=Current, 4=None)";
TITLE2 "Uses CLASS statement to ask SAS to make our group contrasts";
PROC GLM DATA=work.example2;
* CLASS statement lists categorical predictors it will create contrasts for;
* FORMAT asks it to print value labels defined above;
  CLASS sexMW demgroup; FORMAT sexMW Gender. demgroup Dementia.;
* Model Y = fixed effects, options after / print extra output;
  MODEL cognition = age85 grip9 sexMW demgroup / SOLUTION SS3;
* Request group means (age=85, grip=9, have to say is for men), group= F C N;
  ESTIMATE "Intercept for 4None"    intercept 1 sexMW 0 1 demgroup 0 0 1;
  ESTIMATE "Intercept for 2Future"  intercept 1 sexMW 0 1 demgroup 1 0 0;
  ESTIMATE "Intercept for 3Curent"  intercept 1 sexMW 0 1 demgroup 0 1 0;
* Request group differences, group= F C N;
  ESTIMATE "4None vs. 2Future"      demgroup 1 0 -1;
  ESTIMATE "4None vs. 3Current"     demgroup 0 1 -1;
  ESTIMATE "2Future vs. 3Current"   demgroup -1 1 0;
* Need both RUN and QUIT, shut off titles;
RUN; QUIT; TITLE1; TITLE2;

```

**New Solution for Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Intercept	29.26432541 B	0.69850792	41.90	<.0001
age85	-0.40573396	0.11889717	-3.41	0.0007
grip9	0.60422556	0.14977568	4.03	<.0001
sexmw 1Women	-3.65737421 B	0.89143262	-4.10	<.0001
sexmw 2Men	0.00000000 B	.	.	.
demgroup 2Future	-5.72197100 B	1.01907848	-5.61	<.0001
demgroup 3Current	-16.47981327 B	1.52275357	-10.82	<.0001
demgroup 4None	0.00000000 B	.	.	.

**SAS PROC GLM Syntax and Output for Equation 2.9, adding interaction of age\*grip:**

$$\text{Cognition}_i = \beta_0 + \beta_1(\text{Age}_i - 85) + \beta_2(\text{Grip}_i - 9) + \beta_3(\text{SexMW}_i) + \beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) + e_i$$

We can use the model equation to calculate the **simple effect of age** at any level of *grip strength* (age as the effect, grip strength as the moderator):

$$\text{Simple Effect of Age} = \beta_1(\text{Age}_i - 85) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) \rightarrow (\text{Age}_i - 85)[\beta_1 + \beta_6(\text{Grip}_i - 9)]$$

Likewise, we can use the model equation to calculate the **simple effect of grip strength** at any level of *age* (grip strength as the effect, age as the moderator):

$$\text{Simple Effect of Grip} = \beta_2(\text{Grip}_i - 9) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) \rightarrow (\text{Grip}_i - 9)[\beta_2 + \beta_6(\text{Age}_i - 85)]$$

**Examples of these are requested via the ESTIMATE statements...**

```
TITLE1 "Eq 2.9: Model with Age by Grip Interaction";
PROC GLM DATA=work.example2;
* Model Y = fixed effects, options after / print extra output;
  MODEL cognition = age85 grip9 sexMW demNF demNC age85*grip9
    / SOLUTION SS3 INVERSE;
* Request simple effects at other values of age and grip;
  ESTIMATE "Age Effect for Grip=6"          age85 1 age85*grip9 -3;
  ESTIMATE "Age Effect for Grip=9"          age85 1 age85*grip9  0;
  ESTIMATE "Age Effect for Grip=12"         age85 1 age85*grip9  3;

  ESTIMATE "Grip Effect for Age=80"         grip9 1 age85*grip9 -5;
  ESTIMATE "Grip Effect for Age=85"         grip9 1 age85*grip9  0;
  ESTIMATE "Grip Effect for Age=90"         grip9 1 age85*grip9  5;
* Save pieces of output for use in macros (optional);
  ODS OUTPUT FitStatistics=work.Rsq2_9 InvXPX=work.Inv2_9
    OverallANOVA=work.Ano2_9 ParameterEstimates=work.Fix2_9;
```

The INVERSE option is needed for a macro program that calculates regions of significance for the interaction.

The ODS command is saving (in order): R<sup>2</sup>, matrix inverse of XTX, ANOVA R<sup>2</sup> table, and the solution for fixed effects for the macro programs that follow.

```
* Need both RUN and QUIT, shut off title;
RUN; QUIT; TITLE1;
```

**ANOVA Test for Model R<sup>2</sup>**

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	6	19185.04106	3197.50684	36.85	<.0001
Error	543	47111.49712	86.76150		
Corrected Total	549	66296.53818			

R-Square	Coeff Var	Root MSE	cognition Mean
0.289382	37.52580	9.314586	24.82182

**ESTIMATES: Requested Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Age Effect for Grip=6	-0.70301601	0.15336958	-4.58	<.0001
Age Effect for Grip=9	-0.33396058	0.12035656	-2.77	0.0057
Age Effect for Grip=12	0.03509486	0.18715387	0.19	0.8513
Grip Effect for Age=80	0.00432624	0.24733508	0.02	0.9861
Grip Effect for Age=85	0.61941863	0.14874241	4.16	<.0001
Grip Effect for Age=90	1.23451102	0.25540829	4.83	<.0001

**Solution for Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Intercept	29.40780315	0.69490615	42.32	<.0001
age85	-0.33396058	0.12035656	-2.77	0.0057
grip9	0.61941863	0.14874241	4.16	<.0001
sexmw	-3.45563720	0.88727488	-3.89	0.0001
demNF	-5.92254309	1.01363159	-5.84	<.0001
demNC	-16.30040485	1.51254730	-10.78	<.0001
age85*grip9	0.12301848	0.04053626	3.03	0.0025

**Interpret these fixed effects:**

Simple main effect of Age  $\beta_1 =$

Simple main effect of Grip Strength  $\beta_2 =$

Interpret Age by Grip Strength  $\beta_6 \rightarrow$  Age as Simple Effect, Grip as Moderator:

Interpret Age by Grip Strength  $\beta_6 \rightarrow$  Grip as Simple Effect, Age as Moderator:

**Calculate predicted values for “fake people” to plot interaction:**

$$\text{Predicted Cognition}_i = \beta_0 + \beta_1 (\text{Age}_i - 85) + \beta_2 (\text{Grip}_i - 9) + \beta_6 (\text{Age}_i - 85)(\text{Grip}_i - 9)$$

$$\text{Grip Strength} = 12, \text{Age} = 80: 29.41 - 0.33(-5) + 0.62(3) + 0.12(-5)(3) = 31.12$$

$$\text{Grip Strength} = 12, \text{Age} = 85: 29.41 - 0.33(0) + 0.62(3) + 0.12(0)(3) = 31.27$$

$$\text{Grip Strength} = 12, \text{Age} = 90: 29.41 - 0.33(5) + 0.62(3) + 0.12(5)(3) = 31.42$$

$$\text{Grip Strength} = 9, \text{Age} = 80: 29.41 - 0.33(-5) + 0.62(0) + 0.12(-5)(0) = 31.06$$


$$\text{Grip Strength} = 9, \text{Age} = 85: 29.41 - 0.33(0) + 0.62(0) + 0.12(0)(0) = 29.41$$

$$\text{Grip Strength} = 9, \text{Age} = 90: 29.41 - 0.33(5) + 0.62(0) + 0.12(5)(0) = 27.76$$

$$\text{Grip Strength} = 6, \text{Age} = 80: 29.41 - 0.33(-5) + 0.62(-3) + 0.12(-5)(-3) = 31.00$$

$$\text{Grip Strength} = 6, \text{Age} = 85: 29.41 - 0.33(0) + 0.62(-3) + 0.12(0)(-3) = 27.55$$

$$\text{Grip Strength} = 6, \text{Age} = 90: 29.41 - 0.33(5) + 0.62(-3) + 0.12(5)(-3) = 24.10$$



I wrote another macro program to create the **asymptotic covariance matrix**, which has the variances and covariances among the fixed effects, which is called below:

```
/* To Use Asymp Macro, enter:
   DV=           Name of dependent variable
   AnoData=      Name of dataset that stores MSE for model (on ODS)
   InvData=      Name of dataset that stores X'X inv matrix for model (on ODS)
   Npred=        Number of predictors in model */

* Call macro to make asymptotic covariance matrix;
%Asymp(DV=cognition, AnoData=work.Ano2_9, InvData=work.Inv2_9, Npred=6);
```

The asymptotic covariance matrix is needed in order to calculate what are known as **regions of significance**, which describe the values of the moderator as which the simple main effects turn on or off (for significance). I wrote a macro program to do this as well, which is called twice below:

```
/* To use Regions Macro, enter:
   FixData=      Name of dataset that stores fixed effects for model (on ODS)
   InvData=      Name of dataset that stores X'X inv matrix for model (on ODS)
   Pred=         Case-sensitive name of predictor effect regions are for
   Mod=          Case-sensitive name of moderator effect (for region values)
   ModCenter=    Centering point of moderator predictor
   Interact=     Case-sensitive name of interaction effect */

* Call macro to calculate regions of significance for age effect;
%Regions(FixData=work.Fix2_9, InvData=work.Inv2_9, Pred=age85, Mod=grip9, ModCenter=9,
         Interact=age85*grip9);

* Call macro to calculate regions of significance for grip effect;
%Regions(FixData=work.Fix2_9, InvData=work.Inv2_9, Pred=grip9, Mod=age85, ModCenter=85,
         Interact=age85*grip9);
```

**These macro programs then provide the following output for this model:**

The effect of age85 will be significant at centered values of grip9 BELOW the lower bound and ABOVE the upper bound, which translate to these uncentered lower and upper bounds.

Centered	Centered	Uncentered	Uncentered
Lower	Upper	Lower	Upper
0.66813	9.48171	9.66813	18.4817

The effect of grip9 will be significant at centered values of age85 BELOW the lower bound and ABOVE the upper bound, which translate to these uncentered lower and upper bounds.

Centered	Centered	Uncentered	Uncentered
Lower	Upper	Lower	Upper
-14.8144	-2.28566	70.1856	82.7143

To better understand what this is telling us, I requested the simple main effects of each predictor at these lower and upper moderating values by adding to the PROC GLM syntax from before:

```
ESTIMATE "Age Effect for Grip=9.66813"      age85 1 age85*grip9 0.66813;
ESTIMATE "Age Effect for Grip=18.4817"     age85 1 age85*grip9 9.66813;

ESTIMATE "Grip Effect for Age=70.1856"     grip9 1 age85*grip9 -14.8144;
ESTIMATE "Grip Effect for Age=82.7143"     grip9 1 age85*grip9 -2.28566;
```

<b>ESTIMATES: Requested Fixed Effects</b>				
Parameter	Estimate	Standard Error	t Value	Pr >  t
Age Effect for Grip=9.66813	-0.25176824	0.12845343	-1.96	0.0505
Age Effect for Grip=18.4817	0.85539806	0.43199146	1.98	0.0482
Grip Effect for Age=70.1856	-1.20302631	0.61378846	-1.96	0.0505
Grip Effect for Age=82.7143	0.33824021	0.17257180	1.96	0.0505

Accordingly, the effect of age will be significantly negative below grip strength = 9.66 pounds, nonsignificant between 9.66 and 18.48 pounds, and significantly positive after 18.48 pounds.

Likewise, the effect of grip strength will be significantly negative below age = 70.19 years, nonsignificant between age = 70.19 and 82.71 years, and significantly positive after age = 82.71 years.

**SAS PROC GLM Syntax and Output for Equation 2.13, adding 2 manual contrasts for the interaction of sex by dementia group:**

$$\begin{aligned} \text{Cognition}_i = & \beta_0 + \beta_1 (\text{Age}_i - 85) + \beta_2 (\text{Grip}_i - 9) + \beta_3 (\text{SexMW}_i) \\ & + \beta_4 (\text{DemNF}_i) + \beta_5 (\text{DemNC}_i) + \beta_6 (\text{Age}_i - 85)(\text{Grip}_i - 9) \\ & + \beta_7 (\text{SexMW}_i)(\text{DemNF}_i) + \beta_8 (\text{SexMW}_i)(\text{DemNC}_i) + e_i \end{aligned}$$

We can use the model equation to calculate the **sex and dementia group means** for cognition:

	<u>For Men:</u>	<u>For Women:</u>
Cognition for None =	$\beta_0$	$\beta_0 + \beta_3$
Cognition for Future =	$\beta_0 + \beta_4$	$\beta_0 + \beta_4 + \beta_3 + \beta_7$
Cognition for Current =	$\beta_0 + \beta_5$	$\beta_0 + \beta_5 + \beta_3 + \beta_8$

We can determine the **differences between the sex and dementia group means** as follows:

For Men:

None vs. Future =	Future – None =	$(\beta_0 + \beta_4) - (\beta_0)$	= $\beta_4$
None vs. Current =	Current – None =	$(\beta_0 + \beta_5) - (\beta_0)$	= $\beta_5$
Future vs. Current =	Current – Future =	$(\beta_0 + \beta_5) - (\beta_0 + \beta_4)$	= $-\beta_4 + \beta_5$

For Women:

None vs. Future =	Future – None =	$(\beta_0 + \beta_4 + \beta_3 + \beta_7) - (\beta_0 + \beta_3)$	= $\beta_4 + \beta_7$
None vs. Current =	Current – None =	$(\beta_0 + \beta_5 + \beta_3 + \beta_8) - (\beta_0 + \beta_3)$	= $\beta_5 + \beta_8$
Future vs. Current =	Current – Future =	$(\beta_0 + \beta_5 + \beta_3 + \beta_8) - (\beta_0 + \beta_4 + \beta_3 + \beta_7)$	= $-\beta_4 + \beta_5 - \beta_7 + \beta_8$

**These values are then requested via the ESTIMATE statements below...**

```
TITLE1 "Eq 2.13: Add Sex by Dementia Interaction";
PROC GLM DATA=work.example2;
* Model Y = fixed effects, options after / print extra output;
  MODEL cognition = age85 grip9 sexMW demNF demNC age85*grip9 sexMW*demNF sexMW*demNC
    / SOLUTION SS3;
* Save pieces of output for use in macros (optional);
  ODS OUTPUT FitStatistics=work.Rsq2_13;
```



```

* Request group means (hold age=85, grip=9);
ESTIMATE "Intercept: None Men" intercept 1 demNF 0 demNC 0 sexMW 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "Intercept: Future Men" intercept 1 demNF 1 demNC 0 sexMW 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "Intercept: Current Men" intercept 1 demNF 0 demNC 1 sexMW 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "Intercept: None Women" intercept 1 demNF 0 demNC 0 sexMW 1 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "Intercept: Future Women" intercept 1 demNF 1 demNC 0 sexMW 1 sexMW*demNF 1 sexMW*demNC 0;
ESTIMATE "Intercept: Current Women" intercept 1 demNF 0 demNC 1 sexMW 1 sexMW*demNF 0 sexMW*demNC 1;

* Request simple effects of dementia per sex;
ESTIMATE "None vs. Future: Men" demNF 1 demNC 0 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "None vs. Current: Men" demNF 0 demNC 1 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "Future vs. Current: Men" demNF -1 demNC 1 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "None vs. Future: Women" demNF 1 demNC 0 sexMW*demNF 1 sexMW*demNC 0;
ESTIMATE "None vs. Current: Women" demNF 0 demNC 1 sexMW*demNF 0 sexMW*demNC 1;
ESTIMATE "Future vs. Current: Women" demNF -1 demNC 1 sexMW*demNF -1 sexMW*demNC 1;

* Request simple effects of sex per dementia group;
ESTIMATE "Men vs. Women: None" sexMW 1 sexMW*demNF 0 sexMW*demNC 0;
ESTIMATE "Men vs. Women: Future" sexMW 1 sexMW*demNF 1 sexMW*demNC 0;
ESTIMATE "Men vs. Women: Current" sexMW 1 sexMW*demNF 0 sexMW*demNC 1;

* Need both RUN and QUIT, shut off title;
RUN; QUIT; TITLE1;

```

**ANOVA Test for Model R<sup>2</sup>**

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	8	19785.46147	2473.18268	28.77	<.0001
Error	541	46511.07671	85.97242		
Corrected Total	549	66296.53818			


R-Square	Coeff Var	Root MSE	cognition Mean
0.298439	37.35476	9.272131	24.82182

**ESTIMATES: Requested Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Intercept: None Men	29.0701463	0.74849920	38.84	<.0001
Intercept: Future Men	23.0142449	1.49276013	15.42	<.0001
Intercept: Current Men	17.0994158	2.14021810	7.99	<.0001
Intercept: None Women	26.1945523	0.63883397	41.00	<.0001
Intercept: Future Women	20.3029208	1.11863329	18.15	<.0001
Intercept: Current Women	6.3487222	1.94788049	3.26	0.0012
None vs. Future: Men	-6.0559015	1.63512607	-3.70	0.0002
None vs. Current: Men	-11.9707305	2.24495370	-5.33	<.0001
Future vs. Current: Men	-5.9148291	2.58676242	-2.29	0.0226
None vs. Future: Women	-5.8916315	1.27776082	-4.61	<.0001
None vs. Current: Women	-19.8458301	2.02858306	-9.78	<.0001
Future vs. Current: Women	-13.9541986	2.23891711	-6.23	<.0001
Men vs. Women: None	-2.8755941	1.01123720	-2.84	0.0046
Men vs. Women: Future	-2.7113241	1.87406883	-1.45	0.1485
Men vs. Women: Current	-10.7506936	2.89932314	-3.71	0.0002

**Solution for Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Intercept	29.07014634	0.74849920	38.84	<.0001
age85	-0.33479877	0.11988755	-2.79	0.0054
grip9	0.61789286	0.14807943	4.17	<.0001
sexmw	-2.87559408	1.01123720	-2.84	0.0046
demNF	-6.05590147	1.63512607	-3.70	0.0002
demNC	-11.97073055	2.24495370	-5.33	<.0001
age85*grip9	0.12215159	0.04035286	3.03	0.0026
sexmw*demNF	0.16426999	2.07047524	0.08	0.9368
sexmw*demNC	-7.87509954	3.02453647	-2.60	0.0095



	None	Future	Current
Men	29.07	23.01	17.10
Women	26.19	20.29	6.34

The overall model improvement for the  $df=2$  sex\*group interaction is significant according to an  $R^2$  change test:

```
* Call macro to test R2 change from Eq 2_13 vs. Eq 2_9;
%R2Change(N=550, DataL=work.Rsq2_13, NpredL=8, DataS=work.Rsq2_9, NpredS=6);
```

R2 Change F-Test for work.Rsq2\_13 (L) vs. work.Rsq2\_9 (S)

Rsquare L	Rsquare S	Rsquare Diff	Npred L	Npred S	DFnum	DFdenom	Fvalue	Pvalue
0.29844	0.28938	.009056588	8	6	2	541	3.49194	0.031130

**SAS PROC GLM Syntax and Output for Equation 2.13, using the CLASS statement to add the interaction of sex by dementia group:**

```
TITLE1 "Eq 2.13: Add Sex by Dementia Interaction";
PROC GLM DATA=work.example2;
* CLASS statement lists categorical predictors it will create contrasts for;
* FORMAT asks it to print value labels defined above;
CLASS sexMW demgroup; FORMAT sexMW Gender. demgroup Dementia.;
* Model Y = fixed effects, options after / print extra output;
MODEL cognition = age85 grip9 sexMW demgroup age85*grip9 sexMW*demgroup / SOLUTION SS3;
* Save pieces of output for use in macros (optional);
ODS OUTPUT FitStatistics=work.Rsq2_13;
* Request group means (hold age=85, grip=9) -- sex*demgroup= WF WC WN MF MC MN based on CLASS order;
ESTIMATE "Int for 4None 2Men" intercept 1 demgroup 0 0 1 sexMW 0 1 sexMW*demgroup 0 0 0 0 1 0 0 0;
ESTIMATE "Int for 2Future 2Men" intercept 1 demgroup 1 0 0 sexMW 0 1 sexMW*demgroup 0 0 0 1 0 0 0;
ESTIMATE "Int for 3Curent 2Men" intercept 1 demgroup 0 1 0 sexMW 0 1 sexMW*demgroup 0 0 0 0 1 0 0;
ESTIMATE "Int for 4None 1Women" intercept 1 demgroup 0 0 1 sexMW 1 0 sexMW*demgroup 0 0 1 0 0 0 0;
ESTIMATE "Int for 2Future 1Women" intercept 1 demgroup 1 0 0 sexMW 1 0 sexMW*demgroup 1 0 0 0 0 0 0;
ESTIMATE "Int for 3Curent 1Women" intercept 1 demgroup 0 1 0 sexMW 1 0 sexMW*demgroup 0 1 0 0 0 0 0;
* Request simple effects of dementia group per sex -- sex*demgroup= WF WC WN MF MC MN, add /E to check;
ESTIMATE "4None vs. 2Future: 2Men" demgroup 1 0 -1 sexMW*demgroup 0 0 0 1 0 0 -1;
ESTIMATE "4None vs. 3Current: 2Men" demgroup 0 1 -1 sexMW*demgroup 0 0 0 0 1 -1;
ESTIMATE "2Future vs. 3Current: 2Men" demgroup -1 1 0 sexMW*demgroup 0 0 0 -1 1 0;
ESTIMATE "4None vs. 2Future: 1Women" demgroup 1 0 -1 sexMW*demgroup 1 0 -1 0 0 0 0;
ESTIMATE "4None vs. 3Current: 1Women" demgroup 0 1 -1 sexMW*demgroup 0 1 -1 0 0 0 0;
ESTIMATE "2Future vs. 3Current: 1Women" demgroup -1 1 0 sexMW*demgroup -1 1 0 0 0 0 0;
* Request simple effects of sex per dementia group -- sex*demgroup= WF WC WN MF MC MN, add /E to check;
ESTIMATE "2Men vs. 1Women: 4None" sexMW 1 -1 sexMW*demgroup 0 0 1 0 0 -1;
ESTIMATE "2Men vs. 1Women: 2Future" sexMW 1 -1 sexMW*demgroup 1 0 0 -1 0 0 0;
ESTIMATE "2Men vs. 1Women: 3Curent" sexMW 1 -1 sexMW*demgroup 0 1 0 0 -1 0 0;
* Need both RUN and QUIT, shut off title;
RUN; QUIT; TITLE1;
```

**SAS PROC GLM Syntax and Output for Equation 2.18, adding age\*sex, grip\*sex, age\*grip\*sex (no CLASS statement for sex or dementia groups):**

$$\begin{aligned} \text{Cognition}_i = & \beta_0 + \beta_1 (\text{Age}_i - 85) + \beta_2 (\text{Grip}_i - 9) + \beta_3 (\text{SexMW}_i) + \beta_4 (\text{DemNF}_i) + \beta_5 (\text{DemNC}_i) + \beta_6 (\text{Age}_i - 85)(\text{Grip}_i - 9) \\ & + \beta_7 (\text{SexMW}_i)(\text{DemNF}_i) + \beta_8 (\text{SexMW}_i)(\text{DemNC}_i) + \beta_9 (\text{Age}_i - 85)(\text{SexMW}_i) + \beta_{10} (\text{Grip}_i - 9)(\text{SexMW}_i) \\ & + \beta_{11} (\text{Age}_i - 85)(\text{Grip}_i - 9)(\text{SexMW}_i) + e_i \end{aligned}$$

**Simple main effects** can be calculated as:

**Simple main effect** =  
 what it is (main effect)  
 + what modifies it (2-ways)  
 + what modifies it (3-way)

TITLE1 "Eq. 2.18: Model Three-Way Interaction of Sex by Age by Grip";

PROC GLM DATA=work.example2;

\* Model Y = fixed effects, options after / print extra output;

MODEL cognition = age85 grip9 sexMW demNF demNC age85\*grip9 sexMW\*demNF sexMW\*demNC  
 sexMW\*age85 sexMW\*grip9 sexMW\*age85\*grip9 / SOLUTION SS3;

\* Save pieces of output for use in macros (optional);

ODS OUTPUT FitStatistics=work.Rsq2\_18;

\* Request simple effects of Age;  $\beta_1(\text{Age}_i - 85) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) + \beta_9(\text{Age}_i - 85)(\text{SexMW}_i) + \beta_{11}(\text{Age}_i - 85)(\text{Grip}_i - 9)(\text{SexMW}_i)$   
 $= (\text{Age}_i - 85) * [\beta_1 + \beta_6(\text{Grip}_i - 9) + \beta_9(\text{SexMW}_i) + \beta_{11}(\text{Grip}_i - 9)(\text{SexMW}_i)]$

```
ESTIMATE "Age for Grip=6, Men"      age85 1 age85*grip9 -3 sexMW*age85 0 sexMW*age85*grip9 0;
ESTIMATE "Age for Grip=9, Men"     age85 1 age85*grip9 0 sexMW*age85 0 sexMW*age85*grip9 0;
ESTIMATE "Age for Grip=12, Men"    age85 1 age85*grip9 3 sexMW*age85 0 sexMW*age85*grip9 0;

ESTIMATE "Age for Grip=6, Women"   age85 1 age85*grip9 -3 sexMW*age85 1 sexMW*age85*grip9 -3;
ESTIMATE "Age for Grip=9, Women"   age85 1 age85*grip9 0 sexMW*age85 1 sexMW*age85*grip9 0;
ESTIMATE "Age for Grip=12, Women"  age85 1 age85*grip9 3 sexMW*age85 1 sexMW*age85*grip9 3;
```

\* Request simple effects of Grip;  $\beta_2(\text{Grip}_i - 9) + \beta_6(\text{Age}_i - 85)(\text{Grip}_i - 9) + \beta_{10}(\text{Grip}_i - 9)(\text{SexMW}_i) + \beta_{11}(\text{Age}_i - 85)(\text{Grip}_i - 9)(\text{SexMW}_i)$   
 $= (\text{Grip}_i - 9) * [\beta_2 + \beta_6(\text{Age}_i - 85) + \beta_{10}(\text{SexMW}_i) + \beta_{11}(\text{Age}_i - 85)(\text{SexMW}_i)]$

```
ESTIMATE "Grip for Age=80, Men"    grip9 1 age85*grip9 -5 grip9*sexMW 0 sexMW*age85*grip9 0;
ESTIMATE "Grip for Age=85, Men"    grip9 1 age85*grip9 0 grip9*sexMW 0 sexMW*age85*grip9 0;
ESTIMATE "Grip for Age=90, Men"    grip9 1 age85*grip9 5 grip9*sexMW 0 sexMW*age85*grip9 0;

ESTIMATE "Grip for Age=80, Women"  grip9 1 age85*grip9 -5 grip9*sexMW 1 sexMW*age85*grip9 -5;
ESTIMATE "Grip for Age=85, Women"  grip9 1 age85*grip9 0 grip9*sexMW 1 sexMW*age85*grip9 0;
ESTIMATE "Grip for Age=90, Women"  grip9 1 age85*grip9 5 grip9*sexMW 1 sexMW*age85*grip9 5;
```

\* Request simple effects of Sex per Dementia, hold Age=85, Grip=9;  $\beta_3(\text{SexMW}_i) + \beta_7(\text{SexMW}_i)(\text{DemNF}_i) + \beta_8(\text{SexMW}_i)(\text{DemNC}_i) + \beta_9(\text{Age}_i - 85)(\text{SexMW}_i) + \beta_{10}(\text{Grip}_i - 9)(\text{SexMW}_i) + \beta_{11}(\text{Age}_i - 85)(\text{Grip}_i - 9)(\text{SexMW}_i)$   
 $= (\text{SexMW}_i) * [\beta_3 + \beta_7(\text{DemNF}_i) + \beta_8(\text{DemNC}_i) + \beta_9(\text{Age}_i - 85) + \beta_{10}(\text{Grip}_i - 9) + \beta_{11}(\text{Age}_i - 85)(\text{Grip}_i - 9)]$

```
ESTIMATE "Sex for None"            sexMW 1 sexMW*demNF 0 sexMW*demNC 0 sexMW*age85 0 sexMW*grip9 0 sexMW*age85*grip9 0;
ESTIMATE "Sex for Future"          sexMW 1 sexMW*demNF 1 sexMW*demNC 0 sexMW*age85 0 sexMW*grip9 0 sexMW*age85*grip9 0;
ESTIMATE "Sex for Current"         sexMW 1 sexMW*demNF 0 sexMW*demNC 1 sexMW*age85 0 sexMW*grip9 0 sexMW*age85*grip9 0;
```

\* Request simple effects of Dementia per Sex, not conditional on age or grip;  
 Derived from:  $\beta_4(\text{DemNF}_i) + \beta_5(\text{DemNC}_i) + \beta_7(\text{SexMW}_i)(\text{DemNF}_i) + \beta_8(\text{SexMW}_i)(\text{DemNC}_i)$

```
ESTIMATE "None vs. Future: Men"    demNF  1 demNC  0 sexMW*demNF  0 sexMW*demNC  0;
ESTIMATE "None vs. Current: Men"   demNF  0 demNC  1 sexMW*demNF  0 sexMW*demNC  0;
ESTIMATE "Future vs. Current: Men" demNF -1 demNC  1 sexMW*demNF  0 sexMW*demNC  0;

ESTIMATE "None vs. Future: Women"  demNF  1 demNC  0 sexMW*demNF  1 sexMW*demNC  0;
ESTIMATE "None vs. Current: Women" demNF  0 demNC  1 sexMW*demNF  0 sexMW*demNC  1;
ESTIMATE "Future vs. Current: Women" demNF -1 demNC  1 sexMW*demNF -1 sexMW*demNC  1;
```

\* Request simple two-way interactions within Sex\*Dementia Group;  $\beta_7(\text{SexMW}_i)(\text{DemNF}_i) + \beta_8(\text{SexMW}_i)(\text{DemNC}_i)$

```
ESTIMATE "Sex by None vs. Future"  sexMW*demNF  1 sexMW*demNC  0;
ESTIMATE "Sex by None vs. Current" sexMW*demNF  0 sexMW*demNC  1;
ESTIMATE "Sex by Future vs. Current" sexMW*demNF -1 sexMW*demNC  1;
```

\* Request simple two-way interactions of age\*grip;  $\beta_6(\text{Age}_i-85)(\text{Grip}_i-9) + \beta_{11}(\text{Age}_i-85)(\text{Grip}_i-9)(\text{SexMW}_i)$

```
ESTIMATE "Age by Grip: Men"    age85*grip9 1 age85*grip9*sexMW  0;
ESTIMATE "Age by Grip: Women"  age85*grip9 1 age85*grip9*sexMW  1;
```

\* Request simple two-way interactions of sex\*age;  $\beta_9(\text{Age}_i-85)(\text{SexMW}_i) + \beta_{11}(\text{Age}_i-85)(\text{Grip}_i-9)(\text{SexMW}_i)$

```
ESTIMATE "Sex by Age: Grip=6"    sexMW*age85 1 age85*grip9*sexMW -3;
ESTIMATE "Sex by Age: Grip=9"    sexMW*age85 1 age85*grip9*sexMW  0;
ESTIMATE "Sex by Age: Grip=12"   sexMW*age85 1 age85*grip9*sexMW  3;
```

\* Request simple two-way interactions of sex\*grip;  $\beta_{10}(\text{Grip}_i-9)(\text{SexMW}_i) + \beta_{11}(\text{Age}_i-85)(\text{Grip}_i-9)(\text{SexMW}_i)$

```
ESTIMATE "Sex by Grip: Age=80"   sexMW*grip9 1 age85*grip9*sexMW -5;
ESTIMATE "Sex by Grip: Age=85"   sexMW*grip9 1 age85*grip9*sexMW  0;
ESTIMATE "Sex by Grip: Age=90"   sexMW*grip9 1 age85*grip9*sexMW  5;
```

\* Need both RUN and QUIT, shut off title;

RUN; QUIT; TITLE1;

\* Call macro to test R2 change from Eq 2\_18 vs. Eq 2\_13;

%R2Change(N=550, DataL=work.Rsq2\_18, NpredL=11, DataS=work.Rsq2\_13, NpredS=8);

Simple two-way interactions can be calculated using the same logic as simple main effects:

Simple two-way interaction =  
 what it is (2-way)  
 + what modifies it (3-way)

**Fortunately, the overall model did not improve, according to an R<sup>2</sup> change F-test... but output is provided for your reference!**

R2 Change F-Test for work.Rsq2\_18 (L) vs. work.Rsq2\_13 (S)

Rsquare L	Rsquare S	Rsquare Diff	Npred L	Npred S	DFnum	DFdenom	Fvalue	Pvalue
0.30257	0.29844	.004126581	11	8	3	538	1.06108	0.36516

**ANOVA Test for Model R<sup>2</sup>**

Source	DF	Sum of Squares	Mean Square	F Value	Pr > F
Model	11	20059.03951	1823.54905	21.22	<.0001
Error	538	46237.49868	85.94331		
Corrected Total	549	66296.53818			

R-Square	Coeff Var	Root MSE	cognition Mean
0.302565	37.34844	9.270561	24.82182

**ESTIMATES: Requested Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Age for Grip=6, Men	-1.1881602	0.36585649	-3.25	0.0012
Age for Grip=9, Men	-0.4982870	0.20625046	-2.42	0.0160
Age for Grip=12, Men	0.1915861	0.23463810	0.82	0.4146
Age for Grip=6, Women	-0.6182774	0.16936809	-3.65	0.0003
Age for Grip=9, Women	-0.4032146	0.18061946	-2.23	0.0260
Age for Grip=12, Women	-0.1881517	0.31031688	-0.61	0.5446
Grip for Age=80, Men	-0.4105136	0.43682207	-0.94	0.3478
Grip for Age=85, Men	0.7392750	0.23641834	3.13	0.0019
Grip for Age=90, Men	1.8890635	0.45758439	4.13	<.0001
Grip for Age=80, Women	0.1807631	0.32937371	0.55	0.5834
Grip for Age=85, Women	0.5392011	0.19200934	2.81	0.0052
Grip for Age=90, Women	0.8976392	0.36218725	2.48	0.0135
Sex for None	-2.9734897	1.03636525	-2.87	0.0043
Sex for Future	-2.5766063	1.89856249	-1.36	0.1753
Sex for Current	-11.1191918	2.91755246	-3.81	0.0002
None vs. Future: Men	-6.1654275	1.63793371	-3.76	0.0002
None vs. Current: Men	-11.7837134	2.24789958	-5.24	<.0001
Future vs. Current: Men	-5.6182859	2.59176746	-2.17	0.0306
None vs. Future: Women	-5.7685441	1.28133333	-4.50	<.0001
None vs. Current: Women	-19.9294155	2.02920586	-9.82	<.0001
Future vs. Current: Women	-14.1608714	2.24315626	-6.31	<.0001
Sex by None vs. Future	0.3968833	2.07957735	0.19	0.8487
Sex by None vs. Current	-8.1457021	3.02832114	-2.69	0.0074
Sex by Future vs. Current	-8.5425855	3.42768268	-2.49	0.0130
Age by Grip: Men	0.2299577	0.07594863	3.03	0.0026
Age by Grip: Women	0.0716876	0.05760755	1.24	0.2139
Sex by Age: Grip=6	0.5698828	0.40315818	1.41	0.1581
Sex by Age: Grip=9	0.0950725	0.27415806	0.35	0.7289
Sex by Age: Grip=12	-0.3797378	0.38903933	-0.98	0.3295
Sex by Grip: Age=80	0.5912767	0.54708369	1.08	0.2803
Sex by Grip: Age=85	-0.2000738	0.30456726	-0.66	0.5115
Sex by Grip: Age=90	-0.9914243	0.58357783	-1.70	0.0899

**Solution for Fixed Effects**

Parameter	Estimate	Standard Error	t Value	Pr >  t
Intercept	28.95584314	0.79724317	36.32	<.0001
age85	-0.49828704	0.20625046	-2.42	0.0160
grip9	0.73927496	0.23641834	3.13	0.0019
sexmw	-2.97348968	1.03636525	-2.87	0.0043
demNF	-6.16542746	1.63793371	-3.76	0.0002
demNC	-11.78371340	2.24789958	-5.24	<.0001
age85*grip9	0.22995771	0.07594863	3.03	0.0026
sexmw*demNF	0.39688334	2.07957735	0.19	0.8487
sexmw*demNC	-8.14570211	3.02832114	-2.69	0.0074
age85*sexmw	0.09507247	0.27415806	0.35	0.7289
grip9*sexmw	-0.20007382	0.30456726	-0.66	0.5115
age85*grip9*sexmw	-0.15827010	0.09532484	-1.66	0.0974 → This was a close one! ☺

**SAS PROC GLM Syntax and Output for Equation 2.18, adding age\*sex, grip\*sex, age\*grip\*sex (now using CLASS statement for sex or dementia groups):**

```

TITLE1 "Eq. 2.18: Model Three-Way Interaction of Sex by Age by Grip";
PROC GLM DATA=work.example2;
* CLASS statement lists categorical predictors it will create contrasts for;
* FORMAT asks it to print value labels defined above;
  CLASS sexMW demgroup; FORMAT sexMW Gender. demgroup Dementia.;
* Model Y = fixed effects, options after / print extra output;
  MODEL cognition = age85 grip9 sexMW demgroup age85*grip9 sexMW*demgroup sexMW*age85 sexMW*grip9 sexMW*age85*grip9 / SOLUTION SS3;
* Save pieces of output for use in macros (optional);
  ODS OUTPUT FitStatistics=work.Rsq2_18;
* Request simple effects of Age;
  ESTIMATE "Age for Grip=6, 2Men" age85 1 age85*grip9 -3 sexMW*age85 0 1 sexMW*age85*grip9 0 -3;
  ESTIMATE "Age for Grip=9, 2Men" age85 1 age85*grip9 0 sexMW*age85 0 1 sexMW*age85*grip9 0 0;
  ESTIMATE "Age for Grip=12, 2Men" age85 1 age85*grip9 3 sexMW*age85 0 1 sexMW*age85*grip9 0 3;
  ESTIMATE "Age for Grip=6, 1Women" age85 1 age85*grip9 -3 sexMW*age85 1 0 sexMW*age85*grip9 -3 0;
  ESTIMATE "Age for Grip=9, 1Women" age85 1 age85*grip9 0 sexMW*age85 1 0 sexMW*age85*grip9 0 0;
  ESTIMATE "Age for Grip=12, 1Women" age85 1 age85*grip9 3 sexMW*age85 1 0 sexMW*age85*grip9 3 0;
* Request simple effects of Grip;
  ESTIMATE "Grip for Age=80, 2Men" grip9 1 age85*grip9 -5 grip9*sexMW 0 1 sexMW*age85*grip9 0 -5;
  ESTIMATE "Grip for Age=85, 2Men" grip9 1 age85*grip9 0 grip9*sexMW 0 1 sexMW*age85*grip9 0 0;
  ESTIMATE "Grip for Age=90, 2Men" grip9 1 age85*grip9 5 grip9*sexMW 0 1 sexMW*age85*grip9 0 5;
  ESTIMATE "Grip for Age=80, 1Women" grip9 1 age85*grip9 -5 grip9*sexMW 1 0 sexMW*age85*grip9 -5 0;
  ESTIMATE "Grip for Age=85, 1Women" grip9 1 age85*grip9 0 grip9*sexMW 1 0 sexMW*age85*grip9 0 0;
  ESTIMATE "Grip for Age=90, 1Women" grip9 1 age85*grip9 5 grip9*sexMW 1 0 sexMW*age85*grip9 5 0;
* Request simple effects of Sex per Dementia, hold Age=85, Grip=9, sex*demgroup= WF WC WN MF MC MN based on CLASS order, add /E to check;
  ESTIMATE "Sex for 4None" sexMW 1 -1 sexMW*demgroup 0 0 1 0 0 -1 sexMW*age85 0 0 sexMW*grip9 0 0 sexMW*age85*grip9 0 0;
  ESTIMATE "Sex for 2Future" sexMW 1 -1 sexMW*demgroup 1 0 0 -1 0 0 sexMW*age85 0 0 sexMW*grip9 0 0 sexMW*age85*grip9 0 0;
  ESTIMATE "Sex for 3Current" sexMW 1 -1 sexMW*demgroup 0 1 0 0 -1 0 sexMW*age85 0 0 sexMW*grip9 0 0 sexMW*age85*grip9 0 0;
* Request simple effects of Dementia per Sex, not conditional on age or grip, sex*demgroup= WF WC WN MF MC MN, add /E to check;
  ESTIMATE "4None vs. 2Future: 2Men" demgroup 1 0 -1 sexMW*demgroup 0 0 0 1 0 -1;
  ESTIMATE "4None vs. 3Current: 2Men" demgroup 0 1 -1 sexMW*demgroup 0 0 0 0 1 -1;
  ESTIMATE "4Future vs. 3Current: 2Men" demgroup -1 1 0 sexMW*demgroup 0 0 0 -1 1 0;
  ESTIMATE "2None vs. 2Future: 1Women" demgroup 1 0 -1 sexMW*demgroup 1 0 -1 0 0 0;
  ESTIMATE "2None vs. 3Current: 1Women" demgroup 0 1 -1 sexMW*demgroup 0 1 -1 0 0 0;
  ESTIMATE "2Future vs. 3Current: 1Women" demgroup -1 1 0 sexMW*demgroup -1 1 0 0 0 0;
* Request simple two-way interactions within Sex*Dementia Group, sex*demgroup= WF WC WN MF MC MN, add /E to check;
  ESTIMATE "Sex by 4None vs. 2Future" sexMW*demgroup -1 0 1 1 0 -1;
  ESTIMATE "Sex by 4None vs. 3Current" sexMW*demgroup 0 -1 1 0 1 -1;
  ESTIMATE "Sex by 2Future vs. 3Current" sexMW*demgroup 1 -1 0 -1 1 0;
* Request simple two-way interactions of age*grip;
  ESTIMATE "Age by Grip: 2Men" age85*grip9 1 age85*grip9*sexMW 0 1;
  ESTIMATE "Age by Grip: 1Women" age85*grip9 1 age85*grip9*sexMW 1 0;
* Request simple two-way interactions of sex*age;
  ESTIMATE "Sex by Age: Grip=6" sexMW*age85 1 -1 age85*grip9*sexMW -3 3;
  ESTIMATE "Sex by Age: Grip=9" sexMW*age85 1 -1 age85*grip9*sexMW 0 0;
  ESTIMATE "Sex by Age: Grip=12" sexMW*age85 1 -1 age85*grip9*sexMW 3 -3;
* Request simple two-way interactions of sex*grip;
  ESTIMATE "Sex by Grip: Age=80" sexMW*grip9 1 -1 age85*grip9*sexMW -5 5;
  ESTIMATE "Sex by Grip: Age=85" sexMW*grip9 1 -1 age85*grip9*sexMW 0 0;
  ESTIMATE "Sex by Grip: Age=90" sexMW*grip9 1 -1 age85*grip9*sexMW 5 -5;
* Need both RUN and QUIT, shut off title; RUN; QUIT; TITLE1;

```